

graph twoway rspike — Range plot with spikes

[Syntax](#) [Menu](#) [Description](#) [Options](#)
[Remarks and examples](#) [Also see](#)

Syntax

```
twoway rspike y1var y2var xvar [if] [in] [, options]
```

| <i>options</i> | Description |
|-------------------------------------|--|
| vertical | vertical spikes; the default |
| horizontal | horizontal spikes |
| line_options | change look of spike lines |
| axis_choice_options | associate plot with alternative axis |
| twoway_options | titles, legends, axes, added lines and text, by, regions, name, aspect ratio, etc. |

See [G-3] [line_options](#), [G-3] [axis_choice_options](#), and [G-3] [twoway_options](#).

All explicit options are *rightmost*, except [vertical](#) and [horizontal](#), which are *unique*; see [G-4] [concept: repeated options](#).

Menu

Graphics > Twoway graph (scatter, line, etc.)

Description

A range plot has two y variables, such as high and low daily stock price or upper and lower 95% confidence limits.

`twoway rspike` plots a range, using spikes to connect the high and low values.

Also see [G-2] [graph twoway spike](#) for another style of spike chart.

Options

`vertical` and `horizontal` specify whether the high and low y values are to be presented vertically (the default) or horizontally.

In the default `vertical` case, `y1var` and `y2var` record the minimum and maximum (or maximum and minimum) y values to be graphed against each `xvar` value.

If `horizontal` is specified, the values recorded in `y1var` and `y2var` are plotted in the x direction and `xvar` is treated as the y value.

line_options specify the look of the lines used to draw the spikes, including pattern, width, and color; see [G-3] [line_options](#).

axis_choice_options associate the plot with a particular y or x axis on the graph; see [G-3] [axis_choice_options](#).

twoway_options are a set of common options supported by all *twoway* graphs. These options allow you to title graphs, name graphs, control axes and legends, add lines and text, set aspect ratios, create graphs over by() groups, and change some advanced settings. See [G-3] [twoway_options](#).

Remarks and examples

[stata.com](http://www.stata.com)

Remarks are presented under the following headings:

[Typical use](#)

[Advanced use](#)

[Advanced use 2](#)

Typical use


We have daily data recording the values for the S&P 500 in 2001:

```
. use http://www.stata-press.com/data/r13/sp500
(S&P 500)
. list date high low close in 1/5
```

| | date | high | low | close |
|----|-----------|---------|---------|---------|
| 1. | 02jan2001 | 1320.28 | 1276.05 | 1283.27 |
| 2. | 03jan2001 | 1347.76 | 1274.62 | 1347.56 |
| 3. | 04jan2001 | 1350.24 | 1329.14 | 1333.34 |
| 4. | 05jan2001 | 1334.77 | 1294.95 | 1298.35 |
| 5. | 08jan2001 | 1298.35 | 1276.29 | 1295.86 |

We will use the first 57 observations from these data:


```
. twoway rspike high low date in 1/57
```



Advanced use

twoway rspike can be usefully combined with other twoway plottypes (see [G-2] [graph twoway](#)):

```
. twoway rspike high low date, lcolor(gs11) ||  
  line close date || in 1/57
```




We specified `lcolor(gs11)` to tone down the spikes and give the line plot more prominence.

Advanced use 2

A popular financial graph is

```
. use http://www.stata-press.com/data/r13/sp500, clear
(S&P 500)
. replace volume = volume/1000
(248 real changes made)
. twoway
    rspike hi low date ||
    line  close date ||
    bar   volume date, barw(.25) yaxis(2) ||
in 1/57
, ysca(axis(1) r(900 1400))
  ysca(axis(2) r( 9 45))
  ytitle("                Price -- High, Low, Close")
  ytitle(" Volume (millions)", axis(2) astext just(left))
  legend(off)
  subtitle("S&P 500", margin(b+2.5))
  note("Source: Yahoo!Finance and Commodity Systems, Inc.")
```



Also see

- [G-2] [graph twoway rarea](#) — Range plot with area shading
- [G-2] [graph twoway rbar](#) — Range plot with bars
- [G-2] [graph twoway rcap](#) — Range plot with capped spikes
- [G-2] [graph twoway rcapsym](#) — Range plot with spikes capped with marker symbols
- [G-2] [graph twoway rconnected](#) — Range plot with connected lines
- [G-2] [graph twoway rline](#) — Range plot with lines
- [G-2] [graph twoway rscatter](#) — Range plot with markers
- [G-2] [graph twoway rspike](#) — Twoway spike plots